

# Intelligent sensor system maximizes battery life: Interfacing the MSP430F123 Flash MCU, ADS7822, and TPS60311

By Mark Buccini

MSP430 Applications Manager


## Introduction

This article describes how to use event-driven techniques to maintain the lowest possible power consumption in single-cell, battery-powered intelligent sensor applications. These applications detect such things as glass breakage, sound level, fire, and smoke. The same applications include a microcontroller unit (MCU) to control the system and a high-resolution, analog-to-digital converter (ADC) to search periodically for interesting real-time sensor output signals. To extend battery life, system power consumption must be managed to the absolute lowest possible level. To reduce power consumption, intuition points to a slowly operating data acquisition solution. On

the surface this intuitive approach seems reasonable; but it is misleading, especially when analog circuitry is involved. Although power consumption does typically scale with conversion rate, it will be demonstrated that extending system time in standby modes is the largest contributor to total power reduction. Activity must occur as fast as possible in short “bursts,” allowing the system to return quickly to a standby mode. A micropower intelligent sensor system using the MSP430F123 Flash MCU, ADS7822 12-bit ADC, and a single-cell TPS60311 dc/dc boost converter is examined. The entire system is powered from a common 1.5-V AA battery. See Figure 1.

Continued on next page

Figure 1. System solution using MSP430F123, ADS7822, and TPS60311



Continued from previous page

## Enabling single-cell, 3-year battery life

The example examined is powered from a single 1.5-V AA alkaline battery with a typical usable energy value of 1,000 mAh. If the application has a targeted 3-year battery life, total current consumption must be managed as follows:

$$1,000 \text{ mAh}/3 \text{ years}/365 \text{ days}/24 \text{ hours} = 38.0 \text{ } \mu\text{A}.$$

The example presented also assumes that the sensor must be sampled at a 2-ms interval (512 samples per second) to acquire the necessary real-time data to detect an event such as glass breakage.

## Serial interface

The interface between the MSP430F123 MCU and ADS7822 ADC is glueless. The MCU's integrated USART peripheral is used in SPI master mode to drive communication with the ADC. The USART clock (UCLK) is provided to the ADC (DCLOCK), and receives data on the slave out master (SOMI) from the ADC digital output DOUT. In the example, software uses port pin P3.0 to control the ADC chip select ( $\overline{\text{CS}}$ ). Any available port pin can be used. With  $\overline{\text{CS}}$  and software, the MCU directly enables and disables the ADC. The ADS7822 provides 15 bits of data: 2 bits for sampling, 1 null bit, and 12 bits of conversion code. Typically, SPI serial communication exchanges data as 8-bit bytes. To transfer the 15 bits from ADS7822 to the MSP430, two 8-bit transfers are made back-to-back and packed into a single 16-bit word by using software. The word data is rotated right to eliminate the erroneous extra bit and is then AND'ed with 0FFFh to expose only the interesting 12 bits of ADC conversion code. The result is right justified.

The MCU enables the ADC by resetting  $\overline{\text{CS}}$ . The conversion is started and timed with 16 bits of data exchanged by writing two dummy bytes back-to-back to the USART transmit buffer (TXBUF). TXBUF is double buffered, so back-to-back 8-bit writes will transfer a total of 16 bits. Though no data is actually transmitted from the MCU, writing to TXBUF automatically initiates a complete bidirectional SPI transfer, including UCLK generation that

is required by the ADC DCLOCK. The data transmit function of the USART is simply ignored. After the first 8-bit byte of data is received from the ADC into the USART shift register, it is transferred automatically to the receive buffer (RXBUF). The second byte of data is automatically received next as software is recovering the first byte from RXBUF. The second byte of data is recovered; then the 16-bit result is packed into register R15. See Figure 2.

## Slow operating system analysis

First consider the intuitive approach to saving power; that is, simply clock the entire system slow from the available 32-kHz watch crystal. With the MSP430 CPU master clock (MCLK) clocked continuously at 32 kHz, current consumption is approximately 10–12  $\mu\text{A}$ . With this slow clocking, the MCU will be essentially 100% active in order to complete the required sample and decision loop at the 2-ms interval. No standby modes are possible. The SPI UCLK at 1,6384 Hz ( $32,768/2$ ) can conveniently clock the ADC DCLOCK. With 16 clocks per sample, the ADC can complete a conversion in approximately 1 ms, leaving 1 ms for power-down and time for the MCU to complete the calculation. The active current of the ADS7822 is approximately 200  $\mu\text{A}$ . Powered down, the ADC current consumption is a negligible 50 nA. The MCU will be 100% active, and the ADC 50% active.


In snooze mode, quiescent current of the TPS60311 dc/dc charge pump is 2.0  $\mu\text{A}$  typical and can deliver 2 mA to the system. Since the system should never use more than 2 mA, and the 2- $\mu\text{A}$  TPS60311 snooze current will be a constant draw, snooze should be the only mode ever used.

With slow (32-kHz) processing, the system consumes an average current as calculated:

$$\begin{aligned} I_{\text{BAT}} &= \text{MCU} + \text{ADC} + \text{dc/dc} \\ &= (10 \text{ } \mu\text{A} \times 1.00) + (200 \text{ } \mu\text{A} \times 0.50) + (2 \text{ } \mu\text{A}) \\ &= 10 \text{ } \mu\text{A} + 100 \text{ } \mu\text{A} + 2 \text{ } \mu\text{A} \\ &= 112 \text{ } \mu\text{A}. \end{aligned}$$

At an average current consumption of 112  $\mu\text{A}$ , the system exceeds the 38- $\mu\text{A}$  maximum for 3-year, single-cell battery life. The most significant current consumption is the 50%

Figure 2. MSP430F123/ADS7822 serial timing



active duty cycle for the ADC. The analog circuitry of the ADC is the prime current consumer and is largely unaffected by speed of conversion. The ADC's total current consumption is approximately 200  $\mu\text{A}$  and is almost unchanged over the full DCLOCK range of operation at 10 kHz–1.2 MHz. With this in mind, it makes sense to convert as fast as possible and spend as much time as possible in power down. With a slow 32-kHz MCLK, the MCU is 100% active; no standby mode is possible and no processing bandwidth available for increased digital signal processing or additional sensor sampling.

### Burst mode system analysis

Next consider a burst technique—fast, short active-mode intervals with extended system time in standby. The MSP430 is clocked by a single 32-kHz watch crystal that drives an internal auxiliary clock (ACLK). The ACLK is used in the example to drive only the watchdog timer (WDT), configured by software as a 2-ms-interval timer. As configured, the MCU operates normally in low-power mode 3 (LPM3) with only the ACLK and WDT interrupt active. At the programmed 2-ms interval, the WDT interrupts the system and activates the CPU, which is then driven by a high-speed MCLK clocked from the on-chip digitally controlled oscillator (DCO). It is important to note that the fast-starting DCO also allows the MCU to move from standby to high-speed active mode in less than 6  $\mu\text{s}$ . The DCO has been configured by software to operate in the 1-MHz range. The sub-master clock (SMCLK) used by the USART has also been configured to clock from the DCO.

As defined by the programmed WDT interrupt interval, the ADC is sampled and a decision made every 2 ms.

The ADC measurement and decision code requires approximately 70 active CPU MCLK clocks. At 512 measurements/decisions per second, the CPU is active a total of  $70 \times 512 = 35,840$  clocks per second. Assuming that the CPU is clocked at 1 MHz (1  $\mu\text{s}$  per clock), the active duty cycle for the CPU is  $35,840/1,000,000 = 3.5\%$ . The MCU active current is approximately 300  $\mu\text{A}$ . The standby mode LPM3 current is 1.6  $\mu\text{A}$  and is dominant 96.5% of the application time. Using a 500-kHz DCLOCK (1 MHz/2), the system samples the ADS7822 512 times per second, with each sample taking 16 clocks. The ADC will be 1.6% active and 98.4% powered down.

For maximum power conservation, the MCU manages the system such that MCU and ADC spend the vast majority of time in standby mode, with an average standby current consumption of all components  $\sim 3.6 \mu\text{A}$ .

With a burst-mode technique, the system consumes an average current as calculated:

$$\begin{aligned} I_{\text{BAT}} &= \text{MCU} + \text{ADC} + \text{dc/dc} \\ &= (1.6 \mu\text{A} \times 0.965 + 300 \mu\text{A} \times 0.035) \\ &\quad + (200 \mu\text{A} \times 0.016) + (2.0 \mu\text{A}) \\ &= 11.54 \mu\text{A} + 3.2 \mu\text{A} + 2 \mu\text{A} \\ &= 17.2 \mu\text{A}. \end{aligned}$$

With burst-mode programming, the system consumes an average current of only 17.2  $\mu\text{A}$  and easily surpasses the target of 38.0  $\mu\text{A}$  required for 3-year battery life. The additional power reserve can be used to compensate for other requirements such as faster sampling, sensor signal conditioning, or weak/leaky batteries. Also, the current

budgeting does not take into account the power required to support the activity if a trigger event occurs—i.e., glass breakage or fire detection. It is thought that these events would be very rare and clearly the exception to normal operation.

### Further reduction in power consumption

It has been demonstrated that managing the system active duty cycle is of utmost importance for reducing power consumption in intelligent sensing applications. The primary goal should be to minimize active time and maximize time in standby and power-down modes. MCU code execution should be fast and efficient. MCU activity should be event-driven and use interrupts that identify the code to be executed as fast as possible; i.e., avoid cycle-wasting subroutine calls, polling, and bit-test skip chains. The watch crystal and DCO combination is ideal for battery-powered applications, providing both an ultra-low-power standby mode and a fast processor clock. The fact that the DCO can start and is stable in less than 6  $\mu\text{s}$  provides very large power savings by reducing start-up time. The start-up time for a typical crystal or resonator is in the 10-ms range—time that would be wasted and consume significant power. Using the DCO can reduce standby-to-active wake-up time by over 1000 $\times$ .

The ADC must also be powered down as long as possible. If more limited resolution data can be used, consider short cycling. That is, inspect only the first 8 bits of ADC conversion code (which represent the 5 MSB of the conversion) in the first of two SPI transfers required for a complete conversion. Determine if the first 5 bits are interesting; if not, terminate the second SPI transfer to power down the ADC.

In all cases, if a voltage regulator or step-up converter is used, reduce system  $V_{\text{CC}}$  as much as possible. All components will use less power at a lower  $V_{\text{CC}}$ .

### Description of the demonstration program

The code is short, interrupt-driven, and programmed in assembler. This combination ensures that the MCU executes code as fast and effectively as possible with no power-wasting extra cycles.

The demonstration firmware is coded with a Mainloop and single active WDT interrupt service routine (ISR). Three MSP430F123 USART pins on port 3 are used to interface with the ADS7822 ADC, and port 1 pin P1.0 is used as a status indicator. All MSP430 port pins default to the input direction. In a final application, unused pins should be configured to the output direction to avoid power-consuming floating inputs.

The code demonstration firmware first initializes the peripherals. Next, inside the Mainloop, setting the appropriate bits in the CPU status register (SR) forces the processor to enter LPM3:

```
Mainloop    bis.w    #LPM3,SR    ; Enter LPM3
```

The CPU, DCO, and code execution halt at this point, but the pre-configured 2-ms WDT ISR is active with the timer clocked by the 32-kHz ACLK. It is at this point that the code spends the vast majority of time with the MCU and ADC in a battery-conserving standby mode; only the ACLK and WDT are active.

Continued on next page

**Continued from previous page**

As with all MSP430 ISRs, the program counter (PC) and SR are automatically saved on the stack, and the CPU is returned to active. This ensures that the ISR can always be processed, even from a low-power mode. The ISR places the WDT vector on the PC, and code execution begins. Inside the WDT ISR, P3.0 resets  $\overline{CS}$ , enabling the ADS7822. The ADC is sampled and then  $\overline{CS}$  is set, returning the ADC to power-down. For demonstration purposes, if the 12-bit ADC code is greater than 2,048, the `Mainloop` is made active after the WDT ISR by clearing the LPM3 bit on the top of the stack (TOS) that holds the previous SR. Manipulating the PC on the TOS is an excellent technique for using events to manage system activity and low-power modes.

```
bic.w    #LPM3,0(SP) ; Exit LPM3 on reti
```

An active `Mainloop` will toggle port pin P1.0, indicating a trigger to action in a user application. If the ADC code is less than 2,048, the system returns to LPM3. The WDT ISR operates continuously at the programmed 2-ms interval. In an actual end application, a more complex decision/action than a simple comparison/pin toggling could be made when the sensor data is interesting.

**References**

For more information related to this article, you can download an Acrobat Reader file at [www-s.ti.com/sc/techlit/litnumber](http://www-s.ti.com/sc/techlit/litnumber) and replace "litnumber" with the **TI Lit. #** for the materials listed below.

Document Title	TI Lit. #
1. MSP430x1xx Family User's Guide . . . . .	slau049
2. "MSP430x12x Mixed Signal Microcontroller," Data Sheet . . . . .	slas312
3. "12-Bit High-Speed 2.7 V microPower Sampling Analog-to-Digital Converter," ADS7822 Data Sheet . . . . .	sbas062
4. "TPS60310, TPS60311, TPS60312, TPS60313 Single-Cell to 3-V/3.3-V, 20-mA Dual Output, High-Efficiency Charge Pump with Snooze Mode," Data Sheet . . . . .	slvs362

**Related Web sites**

[www.ti.com/sc/docs/products/analog/device.html](http://www.ti.com/sc/docs/products/analog/device.html)

Replace *device* with ads7822, msp430f123 or tps60311

**MSP430F123/ADS7822 demonstration program**

```
#include "msp430x12x.h"
;
;-----
;          ORG          0E000h          ; Program Start
;-----
RESET      mov.w        #0300h,SP          ; Initialize 'x12x stack
SetupWDT   mov.w        #WDT_ADLY_1_9,&WDTCTL ; WDT 1.9ms, ACLK, interval timer
           bis.b        #WDTIE,&IE1        ; Enable WDT interrupt
SetupDCO   mov.b        #0C0h,&DCOCTL      ; DCO~1MHz
SetupP1    bis.b        #01h,&P1DIR        ; P1.0 output direction
SetupP3    bis.b        #0Eh,&P3SEL        ; P3.1-3 SPI option select
           bis.b        #01h,&P3DIR        ; P3.0 output direction
SetupSPI   bis.b        #USPIE0,&ME2       ; Enable USART0 SPI
           bis.b        #CKPH+SSEL1+SSEL0+STC,&UTCTL0 ; SMCLK, 3-pin mode
           mov.b        #CHAR+SYNC+MM,&UCTL0 ; 8-bit SPI Master **clear SWRST**
           mov.b        #02h,&UBR0        ; SMCLK/2 for baud rate
           clr.b        &UBR10           ;
           clr.b        &UMCTL0          ; Clear modulation
           eint              ; Enable interrupts
           ;
Mainloop   bis.w        #LPM3,SR          ; Enter LPM3
ON         bis.b        #001h,&P1OUT      ; P1.0 = 1 - LED ON
OFF        bic.b        #001h,&P1OUT      ; P1.0 = 0 - LED OFF
           jmp          Mainloop          ;
           ;
```

**MSP430F123/ADS7822 demonstration program (Continued)**

```

;-----
WDT_ISR;
;-----
Meas_7822  bic.b    #CS,&P3OUT                ; /CS reset, enable ADC
           mov.b    #00h,&TXBUF0            ; Dummy write to start SPI
           mov.b    #00h,&TXBUF0            ; Dummy write to start SPI
           bic.b    #URXIFG0,&IFG2         ; Clear errant RXBUF flag
RX_MSB     bit.b    #URXIFG0,&IFG2         ; RXBUF ready?
           jnc      RX_MSB                  ; 1 = empty
           mov.b    &RXBUF0,R15            ; R15 = 00|MSB
           swpb     R15                     ; R15 = MSB|00
RX_LSB     bit.b    #URXIFG0,&IFG2         ; RXBUF ready?
           jnc      RX_LSB                  ; 1 = empty
           bis.b    #CS,&P3OUT              ; /CS set, disable ADC
           mov.b    &RXBUF0,R14            ; R14 = 00|LSB
           add.w    R14,R15                 ; R15 = MSB|LSB
           rrc.w    R15                     ;
           and.w    #0FFFh,R15             ; Keep only databits
           cmp.w    #2048,R15              ;
           jlo     No_action                ; jump if 2048 < R15
           bic.w    #LPM3,0(SP)            ; Exit LPM3 on reti
No_action  reti                             ;
;-----
;          Interrupt Vectors Used MSP430F12x
;-----
           ORG      0FFFEh                  ; MSP430 RESET Vector
           DW       RESET                    ;
           ORG      0FFF4h                  ; WDT Vector
           DW       WDT_ISR                  ;
           END

```

## IMPORTANT NOTICE

Texas Instruments Incorporated and its subsidiaries (TI) reserve the right to make corrections, modifications, enhancements, improvements, and other changes to its products and services at any time and to discontinue any product or service without notice. Customers should obtain the latest relevant information before placing orders and should verify that such information is current and complete. All products are sold subject to TI's terms and conditions of sale supplied at the time of order acknowledgment.

TI warrants performance of its hardware products to the specifications applicable at the time of sale in accordance with TI's standard warranty. Testing and other quality control techniques are used to the extent TI deems necessary to support this warranty. Except where mandated by government requirements, testing of all parameters of each product is not necessarily performed.

TI assumes no liability for applications assistance or customer product design. Customers are responsible for their products and applications using TI components. To minimize the risks associated with customer products and applications, customers should provide adequate design and operating safeguards.

TI does not warrant or represent that any license, either express or implied, is granted under any TI patent right, copyright, mask work right, or other TI intellectual property right relating to any combination, machine, or process in which TI products or services are used. Information published by TI regarding third-party products or services does not constitute a license from TI to use such products or services or a warranty or endorsement thereof. Use of such information may require a license from a third party under the patents or other intellectual property of the third party, or a license from TI under the patents or other intellectual property of TI.

Reproduction of information in TI data books or data sheets is permissible only if reproduction is without alteration and is accompanied by all associated warranties, conditions, limitations, and notices. Reproduction of this information with alteration is an unfair and deceptive business practice. TI is not responsible or liable for such altered documentation.

Resale of TI products or services with statements different from or beyond the parameters stated by TI for that product or service voids all express and any implied warranties for the associated TI product or service and is an unfair and deceptive business practice. TI is not responsible or liable for any such statements.

Following are URLs where you can obtain information on other Texas Instruments products and application solutions:

### Products

Amplifiers	<a href="http://amplifier.ti.com">amplifier.ti.com</a>
Data Converters	<a href="http://dataconverter.ti.com">dataconverter.ti.com</a>
DSP	<a href="http://dsp.ti.com">dsp.ti.com</a>
Interface	<a href="http://interface.ti.com">interface.ti.com</a>
Logic	<a href="http://logic.ti.com">logic.ti.com</a>
Power Mgmt	<a href="http://power.ti.com">power.ti.com</a>
Microcontrollers	<a href="http://microcontroller.ti.com">microcontroller.ti.com</a>

### Applications

Audio	<a href="http://www.ti.com/audio">www.ti.com/audio</a>
Automotive	<a href="http://www.ti.com/automotive">www.ti.com/automotive</a>
Broadband	<a href="http://www.ti.com/broadband">www.ti.com/broadband</a>
Digital control	<a href="http://www.ti.com/digitalcontrol">www.ti.com/digitalcontrol</a>
Military	<a href="http://www.ti.com/military">www.ti.com/military</a>
Optical Networking	<a href="http://www.ti.com/opticalnetwork">www.ti.com/opticalnetwork</a>
Security	<a href="http://www.ti.com/security">www.ti.com/security</a>
Telephony	<a href="http://www.ti.com/telephony">www.ti.com/telephony</a>
Video & Imaging	<a href="http://www.ti.com/video">www.ti.com/video</a>
Wireless	<a href="http://www.ti.com/wireless">www.ti.com/wireless</a>

## TI Worldwide Technical Support

### Internet

**TI Semiconductor Product Information Center Home Page**  
[support.ti.com](http://support.ti.com)

**TI Semiconductor KnowledgeBase Home Page**  
[support.ti.com/sc/knowledgebase](http://support.ti.com/sc/knowledgebase)

### Product Information Centers

#### Americas

Phone	+1(972) 644-5580	Fax	+1(972) 927-6377
Internet/Email	<a href="http://support.ti.com/sc/pic/americas.htm">support.ti.com/sc/pic/americas.htm</a>		

#### Europe, Middle East, and Africa

Phone			
Belgium (English)	+32 (0) 27 45 54 32	Netherlands (English)	+31 (0) 546 87 95 45
Finland (English)	+358 (0) 9 25173948	Russia	+7 (0) 95 7850415
France	+33 (0) 1 30 70 11 64	Spain	+34 902 35 40 28
Germany	+49 (0) 8161 80 33 11	Sweden (English)	+46 (0) 8587 555 22
Israel (English)	1800 949 0107	United Kingdom	+44 (0) 1604 66 33 99
Italy	800 79 11 37		
Fax	+(49) (0) 8161 80 2045		
Internet	<a href="http://support.ti.com/sc/pic/euro.htm">support.ti.com/sc/pic/euro.htm</a>		

#### Japan

Fax			
International	+81-3-3344-5317	Domestic	0120-81-0036
Internet/Email			
International	<a href="http://support.ti.com/sc/pic/japan.htm">support.ti.com/sc/pic/japan.htm</a>		
Domestic	<a href="http://www.tij.co.jp/pic">www.tij.co.jp/pic</a>		

#### Asia

Phone			
International	+886-2-23786800		
Domestic	Toll-Free Number		
Australia	1-800-999-084	New Zealand	0800-446-934
China	800-820-8682	Philippines	1-800-765-7404
Hong Kong	800-96-5941	Singapore	800-886-1028
Indonesia	001-803-8861-1006	Taiwan	0800-006800
Korea	080-551-2804	Thailand	001-800-886-0010
Malaysia	1-800-80-3973		
Fax	886-2-2378-6808	Email	<a href="mailto:tiasia@ti.com">tiasia@ti.com</a>
Internet	<a href="http://support.ti.com/sc/pic/asia.htm">support.ti.com/sc/pic/asia.htm</a>		<a href="mailto:ti-china@ti.com">ti-china@ti.com</a>

### C011905

**Safe Harbor Statement:** This publication may contain forward-looking statements that involve a number of risks and uncertainties. These "forward-looking statements" are intended to qualify for the safe harbor from liability established by the Private Securities Litigation Reform Act of 1995. These forward-looking statements generally can be identified by phrases such as "TI or its management believes," "expects," "anticipates," "foresees," "forecasts," "estimates" or other words or phrases of similar import. Similarly, such statements herein that describe the company's products, business strategy, outlook, objectives, plans, intentions or goals also are forward-looking statements. All such forward-looking statements are subject to certain risks and uncertainties that could cause actual results to differ materially from those in forward-looking statements. Please refer to TI's most recent Form 10-K for more information on the risks and uncertainties that could materially affect future results of operations. We disclaim any intention or obligation to update any forward-looking statements as a result of developments occurring after the date of this publication.

**Trademarks:** All trademarks are the property of their respective owners.

Mailing Address: Texas Instruments  
Post Office Box 655303  
Dallas, Texas 75265

© 2005 Texas Instruments Incorporated